Problem:
Score : 300 points

Problem Statement

You drew lottery N times. In the i-th draw, you got an item of the kind represented by a string S_i.

How many kinds of items did you get?

Constraints

1 \leq N \leq 2\times 10^5

S_i consists of lowercase English letters and has a length between 1 and 10 (inclusive).

Input

Input is given from Standard Input in the following format:

N
S_1
:
S_N

Output

Print the number of kinds of items you got.

Sample Input 1

3
apple
orange
apple

Sample Output 1

2

You got two kinds of items: apple and orange.

Sample Input 2

5
grape
grape
grape
grape
grape

Sample Output 2

1

Sample Input 3

4
aaaa
a
aaa
aa

Sample Output 3

4

Solution in Pascal (FPC 3.0.4) (judge verdict: Accepted)

program hi;
type ars = array[1..200000] of string;
var 
    n, i, cnt: longint;
    a, tmp: ars;

procedure merge(var a, tmp: ars; l, m, r: longint);
var i, i1, i2: longint;
begin
    i := l; i1 := l; i2 := m + 1;
    while (i1 <= m) and (i2 <= r) do begin
        if a[i1] <= a[i2] then begin
            tmp[i] := a[i1];
            inc(i1);
        end else begin
            tmp[i] := a[i2];
            inc(i2);
        end;
        inc(i);
    end;
    if i1 <= m then 
        while i1 <= m do begin
            tmp[i] := a[i1];
            inc(i1); inc(i);
        end 
    else 
        while i2 <= r do begin
            tmp[i] := a[i2];
            inc(i2); inc(i);
        end;
    for i := l to r do 
        a[i] := tmp[i];
end;

procedure mSort(var a, tmp: ars; l, r: longint);
var m: longint;
begin
    if l < r then begin
        m := (l + r) div 2;
        mSort(a, tmp, l, m);
        mSort(a, tmp, m + 1, r);
        merge(a, tmp, l, m, r);
    end;
end;

begin
    readln(n);
    for i := 1 to n do 
        readln(a[i]);
    mSort(a, tmp, 1, n);
    cnt := 1;
    for i := 1 to n - 1 do 
        if a[i] <> a[i + 1] then
            inc(cnt);
    writeln(cnt);
end.